

# Lokalisatie met Arduino's op basis van vier bakens

Peter van Dijk & Elizabeth Schermerhorn

8 november 2014

# Inhoudsopgave

<b>1</b>	<b>Inleiding</b>	<b>3</b>
<b>2</b>	<b>Probleemstelling</b>	<b>3</b>
<b>3</b>	<b>Gerelateerd werk</b>	<b>4</b>
<b>4</b>	<b>Instellingen</b>	<b>4</b>
4.1	De hardware . . . . .	5
4.1.1	Ontvanger . . . . .	5
4.1.2	zender . . . . .	5
4.2	De software . . . . .	6
<b>5</b>	<b>Het algoritme</b>	<b>7</b>
5.1	Afstand bepalen tot aan de bakens . . . . .	7
5.2	Trilateratie . . . . .	8
<b>6</b>	<b>Testopstelling en metingen</b>	<b>9</b>
<b>7</b>	<b>Conclusie</b>	<b>10</b>
7.1	Aanbevelingen . . . . .	10
<b>A</b>	<b>Code Trilateration</b>	<b>13</b>
<b>B</b>	<b>Meetresultaten</b>	<b>15</b>

# 1 Inleiding

Dit is het verslag voor de opdracht 'Lokalisatie' van het vak Networked Smart Systems. Voor deze opdracht moest er een lokalisatiealgoritme worden ontwikkeld waarmee binnen een vak van 5 bij 5 meter de positie zo nauwkeurig mogelijk proberen te bepalen. Hierbij wordt gebruik gemaakt van ultrasone signalen vanuit vaste referentiepunten op de hoeken van dit vak. De locatie van een meetinstrument kan van belang zijn bij het verrichtingen van metingen, bijvoorbeeld bij een branddetector in een bos.

In dit paper zal het door ons geïmplementeerde lokalisatiealgoritme besproken worden. Er zal eerst een afweging worden gemaakt tussen bestaande algoritmen. Daarna zal de gebruikte hardware en software voor de lokalisatie besproken worden. Vervolgens wordt er uit deze algoritmen het algoritme gekozen dat het beste aansluit op deze hardware en software en de implementatie hiervan uitgelegd worden. Daaropvolgend zullen de testopstelling en de resultaten worden uitgelicht. Ten slotte zal hieruit een conclusie worden getrokken en zullen er aanbevelingen voor vervolgonderzoek gedaan worden.

# 2 Probleemstelling

Aan de hand van vier bakens die in een 5 bij 5 meter veld geplaatst worden, moet de positie van de ontvanger bepaald worden. De onderzoeksvraag die hier centraal staat: *“Is het mogelijk om met een afwijking kleiner dan 5 procent de positie van de ontvanger in een 5 bij 5 meter veld te bepalen?”* De verwachte uitkomst van dit onderzoek is dat het eindproduct in staat is om met een nauwkeurigheid van 25 centimeter (5 procent) de positie te bepalen binnen het 5 bij 5 meter veld.

Een limitatie bij deze positiebepaling is onder andere de gebruikte hardware. Er wordt gebruik gemaakt van een Arduino Uno, een computer met beperkte rekenkracht en beperkt werkgeheugen. Verwachte obstakels hierdoor zijn bit overflows en het beperkte aantal variabelen in de code (in verband met het beperkte werkgeheugen). De overflow wordt als probleem verwacht omdat de Arduino geen grote getallen kan representeren en er waarschijnlijk met grote getallen gerekend moet gaan worden. Een voorbeeld hiervan is bijvoorbeeld de halvering van het bereik van getallen. Zo is een `long` geen 64, maar 32 bits.[1] Een ander verwacht obstakel is de snelheid van het geluidssignaal dat de bakens zullen gaan uitzenden, omdat de geluidssnelheid varieert naarmate de temperatuur en luchtvochtigheid veranderen. Dit zou gevolgen kunnen hebben voor de te bepalen positie.

### 3 Gerelateerd werk

Op dit gebied is er al veel onderzoek verricht. Hierdoor is er veel werk wat gebruikt kan worden bij het onderzoek. Aangezien dit een onderzoek is wat al vele malen is uitgevoerd is er veel gerelateerd werk dat is gebruikt in dit onderzoek. In deze twee papers worden er globale lokalisatie systemen (GPS) beschreven en hoe deze gebruik maken van externe factoren. In de volgende twee papers worden er trilateratiealgoritmen gepresenteerd waarmee de exacte locatie bepaald kan worden. In de eerste twee papers worden twee manieren toegelicht waarop lokalisering mogelijk is. In *A New Location Technique for the Active Office* worden twee manieren beschreven [4], namelijk:

- Active badges
- ParcTab

Active Badges is gebaseerd op het periodiek verzenden van informatie naar iedereen. Door bijvoorbeeld een gebouw heen zijn er overal sensoren geplaatst die luisteren naar berichten die worden verstuurd door badges. Door vast te stellen welke sensoren het signaal van welke badge hebben ontvangen is het mogelijk een grove schatting te geven van de locatie van de badge.


ParcTab is een zakcomputer die gebruik maakt van een netwerk dat communiceert via infrarood signalen. Wat betreft de lokalisatie lijkt deze heel veel op de badges die hiervoor staan beschreven. Er worden berichten verstuurd en afhankelijk van welke sensor deze ontvangen wordt de locatie bepaald. Een groot verschil is dat de ParcTab is bedoeld voor intensiever gebruik dan alleen het berekenen van de locatie.

In het algoritme in het hoofdstuk *Localization and positioning* van *Protocols and Architectures for Wireless Sensor Networks* wordt er gebruik gemaakt van bakens die een signaal uitzenden [3]. Met behulp van deze bakens kan op twee manieren een positie worden berekend: met *triangulatie* (Figuur 2) en *trilateratie* (Figuur 1). Triangulatie maakt gebruik van de invalshoek om de positie te berekenen. Hierbij moeten de hoeken tot tenminste twee bakens en de afstand tussen de bakens bekend zijn. Trilateratie gebruikt de afstand tot de bakens om de positie te bepalen. Voor dit algoritme is het van belang dat er minimaal tot drie bakens de afstand bekend is. Wanneer de afstand tot drie bakens bekend is kan met een simpele matrixmultiplicatie de positie berekend worden. Dit is het algoritme dat ook is gebruikt in de in dit paper besproken oplossing.


### 4 Instellingen

Om deze hypothesen te toetsen moeten er metingen gedaan worden. Hier zijn verschillende zaken van belang.

- De gebruikte hardware



**Figuur 1: Trilateratie [3]**



**Figuur 2: Triangulatie [3]**

- De gebruikte software
- De instellingen/vastgestelde constanten
- De onderzoekopstelling


## 4.1 De hardware

### 4.1.1 Ontvanger


De Hardware die gebruikt is voor het ontvangen van de signalen is een Arduino UNO en een Nordic nRF24L01 draadloze transceiver in combinatie met een ultrasoonontvanger. In Figuur 3 en 4 staat de gebruikte ontvanger vanuit twee posities bekeken.

### 4.1.2 zender

Als zender is er gebruik gemaakt van vier bakens die een ultrasoon geluid uitzenden. Het signaal van deze bakens wordt verstuurd in de richting waarin ze wijzen. Deze bakens bevatten hardware die vergelijkbaar is met die van de ontvanger, met als grote verschil dat er in plaats van een ultrasoonontvanger een ultrasoonzender is gebruikt.



**Figuur 3: De ontvanger (voorkant)**



**Figuur 4: De ontvanger (achterkant)**

## 4.2 De software

Om de RF24 radio aan te sturen wordt gebruik gemaakt van de opensource-softwarebibliotheek voor de Arduino [2]. Met deze software kan de radio aangestuurd worden en kunnen pakketjes worden verzonden en ontvangen. De radio luistert en verstuurt pakketten over een bepaald frequentiekanaal. Het is van belang dat de radio's hetzelfde frequentiekanaal gebruiken, zodat ze met elkaar kunnen communiceren. De kanalen hebben een nummer van 0 tot en met 125. De gebruikte instellingen zijn als volgt:


- Frequentiekanaal: 76
- Automatisch herzenden: uit
- Transmissiesnelheid: 2 Mbps
- Adres pipe: 0xdeadbeef1
- Payload-grootte: 1 byte

De ultrasoonontvanger gebruikt de volgende instellingen:

- 40 kHz signaal: uit
- Versterking: 50x

Om de tests te kunnen uitvoeren is er gebruik gemaakt van vier bakens die met een bepaald patroon informatie verzenden. In Figuur 5 is schematisch

weergegeven hoe de signaalverzending werkt. In het schema zijn vier dezelfde iteraties te zien. Omdat er gebruik wordt gemaakt van vier bakens staan er vier berichten in het schema weergegeven.



Figuur 5: Het signaal van de bakens

## 5 Het algoritme

Het algoritme dat geïmplementeerd is, bestaat uit twee verschillende onderdelen. Het eerste deel bestaat uit het berekenen van de afstand tot de bakens en het tweede deel uit de berekening van de positie. Deze twee onderdelen zullen apart besproken worden voor de overzichtelijkheid.

### 5.1 Afstand bepalen tot aan de bakens

Zoals te zien is in Figuur5 wordt er eerst een radiosignaal verstuurd, gevolgd door een ultrasoon signaal. De bakens gebruiken een master-slave opstelling, waarbij de master aangeeft wanneer elke slave mag sturen. Als eerste wordt er door de master-baken een radio uitzending uitgevoerd met als bericht het ID van de baken die vervolgens mag zenden. Omdat dit een radiosignaal is gaat dit met de snelheid van het licht, hierdoor is het tijdsverschil tussen de verschillende nodes die het op verschillende momenten ontvangen verwaarloosbaar. Op het moment dat het radiosignaal ontvangen wordt, begint de desbetreffende baken direct met het zenden van een ultrasoon geluid. Er is bekend dat dit meteen gebeurt, dus door bij te houden hoeveel tijd verstrijkt tussen het ontvangen van het radiosignaal en het ontvangen van het ultrasoon geluid kan er een schatting gemaakt worden van de afstand tussen baken en ontvanger door de geluidssnelheid te gebruiken. Deze berekening ziet er als volgt uit:

$$d_b = (T_u - T_r) * g$$

Met:

$d_b$ : afstand tot baken  
 $T_u$ : tijd ontvangen ultrasoon geluid  
 $T_r$ : tijd ontvangen radio  
 $g$ : geluidssnelheid

Het is van belang om een juiste geluidssnelheid te kiezen bij de omgeving waarin het systeem staat, de afwijkingen kunnen groot zijn. Dit proces van luisteren en berekenen herhaalt zich oneindig lang zodat wanneer de node in beweging is er nog steeds een correcte locatie berekend kan worden.

## 5.2 Trilateratie

Nu van alle vier de bakens bekend is wat hun afstand is tot de node kan door middel van matrixberekeningen de positie worden bepaald. Hierbij is er gebruik gemaakt van [iets]. Afstanden zijn gegeven in centimeter, tevens zijn coördinaten weergegeven in centimeters afstand tot het nulpunt. De volgende berekening is gebruikt bij het berekenen van de locatie:

$$2 * \begin{bmatrix} x_3 - x_1 & y_3 - y_1 \\ x_3 - x_2 & y_3 - y_2 \end{bmatrix} \begin{bmatrix} x_u \\ y_u \end{bmatrix} = \begin{bmatrix} (r_1^2 - r_3^2) - (x_1^2 - x_3^2) - (y_1^2 - y_3^2) \\ (r_2^2 - r_3^2) - (x_2^2 - x_3^2) - (y_2^2 - y_3^2) \end{bmatrix}$$

Met:

$x_u$ : de te berekenen x-coördinaat van de Arduino.  
 $y_u$ : de te berekenen y-coördinaat van de Arduino.  
 $x_i$ : het x-coördinaat van baken  $i$ .  
 $y_i$ : het y-coördinaat van baken  $i$ .  
 $r_i$ : de afstand van de Arduino tot aan baken  $i$ .

Deze matrixmultiplicatie is om te schrijven naar de volgende formules, waarmee de x- en y-coördinaten van de ontvanger te berekenen zijn:

$$c_1 = (((r_1 * r_1) - (r_3 * r_3)) - ((x_1 * x_1) - (x_3 * x_3)) - ((y_1 * y_1) - (y_3 * y_3)))/2$$

$$c_2 = (((r_2 * r_2) - (r_3 * r_3)) - ((x_2 * x_2) - (x_3 * x_3)) - ((y_2 * y_2) - (y_3 * y_3)))/2$$

$$n = -((y_3 - y_1) * 10000)/(x_3 - x_1)$$

$$m = ((c_1 * 10000)/(x_3 - x_1))$$

$$y_u = ((c_2 * 10000) - m * (x_3 - x_2))/(n * (x_3 - x_2) + (10000 * (y_3 - y_2)))$$

$$x_u = ((n * y) + m)/10000$$




Wat opvalt aan deze berekening is dat er een aantal maal 10.000 in staat. Deze factor is geïntroduceerd om te voorkomen dat afkapping van getallen zorgt voor foute resultaten. Als deze factor niet gebruikt zou worden dan zouden tussenberekeningen een getal kleiner dan 1 kunnen opleveren, wat afgekapt wordt tot 0. Hiermee doorrekenen levert incorrecte resultaten op. De factor wordt aan het einde weer weggedeeld, om het gewenste resultaat te verkrijgen. Een alternatief voor deze oplossing is het gebruik van decimale getallen, maar het rekenen hiermee is erg langzaam, daarom is daar niet voor gekozen.

Doordat de afstand tot aan de bakens elke 400 milliseconden opnieuw wordt berekend geldt dit ook voor de x- en y-coördinaten. Voor elke vier nieuwe afstandsmetingen worden de x- en y-coördinaten opnieuw berekend. Het algoritme maakt gebruik van alle vier de bakens en niet slechts drie voor de positiebepaling. Dit wordt gedaan als volgt. Voor elke combinatie van drie bakens wordt met behulp van de afstanden tot deze bakens met behulp van het hiervoor genoemde algoritme een set van coördinaten berekend. Vervolgens wordt het gemiddelde genomen van de berekende sets coördinaten. Dit zorgt er voor dat er voor kleine afwijkingen in afstandsberekening tot bakens gecompenseerd kan worden.

## 6 Testopstelling en metingen

Om de implementatie te testen is er gebruik gemaakt van een testopstelling zoals die in Figuur 6 staat weergegeven. De unisoontvangers waarmee de signalen



**Figuur 6: Meetopstelling**

van de bakens worden ontvangen werden op een afstand groter dan vijf meter zeer onnauwkeurig dus is ervoor gekozen om binnen een veld van vijf bij vijf meter te werken. In het schema staan de vier bakens aangegeven met de oranje stippen samen met hun coördinaten. Met de roze stippen worden de testlocaties bedoeld waarop is getest. De bakens die zijn gebruikt versturen alleen signalen in de richting waarin ze wijzen. Hierdoor is het niet mogelijk correcte gegevens te verkrijgen wanneer er vanachter de bakens wordt gemeten.

Tijdens het testen wordt de ontvanger op een bepaald coördinaat geplaatst. Het is belangrijk dat deze locatie precies is, anders ontstaat er een afwijking tussen de meetresultaten en de verwachte waarde. Het is van belang bij het testen dat er gebruik wordt gemaakt van een "schoon" grid. Dit houdt in, geen obstakels die de metingen zouden kunnen verstoren. Alle tests zijn "schoon" bepaald. Op de volgende locaties zijn de volgende tests uitgevoerd:

Test 1: De Arduino wordt op het coördinaat (290, 434) geplaatst.

Test 2: De Arduino wordt op het coördinaat (60, 155) geplaatst.

Test 3: De Arduino wordt op het coördinaat (72, 377) geplaatst.

Verwacht wordt dat de metingen zeer nauwkeurig zullen zijn met een afwijking kleiner dan tien centimeter. In Appendix B staan alle meetresultaten van de drie tests verwerkt. In de tabellen staat de laatste waarde voor het gemiddelde dat werd berekend door de Arduino. de overige waarden zijn 25 gemeten waarden op die positie. Te zien is dat de metingen niet hetzelfde zijn, ze fluctueren binnen een interval. Een paar redenen hiervoor zijn:

- De ontvanger is bewogen tijdens de meting.
- De geluidssnelheid is erg belangrijk voor de berekening en is niet correct afgestemd op de temperatuur en luchtvochtigheid van de ruimte.
- De ultrasoonontvanger ontvangt echo's die weerkaatsen op objecten. Deze foutieve metingen worden meegenomen in de berekening waardoor de waarde niet meer klopt.

## 7 Conclusie

De onderzoeksvraag die aan het begin gesteld werd: *"Een algoritme opstellen waarmee de positie bepaald kan worden met een afwijking van minder dan tien centimeter."* Met dit experiment is het gelukt een lokalisatiealgoritme te ontwerpen en te implementeren met een afwijking kleiner dan tien centimeter. Uit de onderzoeksresultaten die weergegeven staan in Appendix B kan dit geconcludeerd worden. In Appendix B staat in de onderste tabel van elke grafiek de gemiddelde afwijking in centimeters. Voor bovenstaande hypothese geldt wel dat de ontvanger zich vóór de bakens moet bevinden en binnen een veld van vijf bij vijf meter. Anders kan er niet gegarandeerd worden dat de afwijking kleiner dan tien centimeter is.

### 7.1 Aanbevelingen

Voor een vervolgonderzoek zou er naar de volgend punten gekeken kunnen worden.

- de overflow die ontstaat met het rekenen met integers, op het moment dat de Arduino verder staat dan zes meter bij de bakens vandaan ontstaat er overflow.

- betere bakens en ontvangers, de signalen worden minder na 6 meter
- een algoritme ontwerpen wat beter tegen verstoringen in het veld kan.

## Referenties

- [1] Arduino Reference: long. <http://arduino.cc/en/Reference/long>, Nov. 2014.
- [2] RF24 Class Reference. <http://maniacbug.github.io/RF24/classRF24.html>, May 2014.
- [3] H. Karl and A. Willig. *Protocols and Architectures for Wireless Sensor Networks*. John Wiley & Sons, 2005.
- [4] A. Ward, A. Jones, and A. Hopper. A new location technique for the active office. *IEEE Personal Communications*, 4(5):42–47, 1997.

## A Code Trilateration

```
#include <SPI.h>
#include "nRF24L01.h"
#include "RF24.h"
#include "printf.h"

RF24 radio(3, 9);
const int audioPin = 1;
const uint64_t pipes[1] = { 0xdeadbeef1LL };

int32_t x[4] = {100,200,300,400};
int32_t y[4] = {20,0,0,20};

uint8_t radio_received = 0;    //Received index of beacon
unsigned long radio_stopped;    //t0
unsigned long audio_started;    //t1
uint32_t distance;              //(t1-t0)*speed_of_sound
unsigned long speed_of_sound = 34421; //344.21 m/s = 34421 / 1000000 cm/us
int offset = 7;                 //7 cm

uint32_t distances[4];          //calculated distances
int32_t x_crds[8];              //calculated coords
int32_t y_crds[8];              //calculated coords

int counter = 0;
int count = 0;
boolean bereken = false;

void setup(void)
{
  Serial.begin(57600);
  printf_begin();
  radio.begin();
  radio.setChannel(76);
  radio.setRetries(0,0);
  radio.setPayloadSize(1);
  radio.setDataRate(RF24_2MBPS);
  radio.openReadingPipe(1,pipes[0]);
  radio.startListening();
  radio.setAutoAck(false);
  radio.printDetails();
}

void loop(void)
{
  if ( radio.available() )
  {
    bool done = false;
    while (!done)
    {
      done = radio.read( &radio_received, sizeof(uint8_t) );
    }
    bool received = false;
    bool waiting = true;
    radio_stopped = micros();
    while(waiting){
      if(analogRead(audioPin)>0){//radio signal received
        audio_started = micros();
        received = true;
        waiting = false;
      }
      else if(micros()-radio_stopped > 75000){//nothing received after radio
        waiting = false;
      }
    }

    if(received){
```

```

        distance = (audio_started-radio_stopped)*speed_of_sound/1000000;
        distance = distance -offset;
        if(distance< 700){//No false readings
            if(distance >= 0) {
                distances[radio_received] = distance;
            }
            else {
                distances[radio_received] = 0;
            }
        }
    }
    if(count == 3){
        for(int i=0;i<4;i++){
            printf("%i : %ld cm ", i, distances[i]);
        }
        printf("\n\r");
        counter = 0;
        while(counter < 4){
            calculateXY((counter%4) , ((counter+1)%4), ((counter+2)%4));
            counter++;
        }
        if(bereken){
            printGemiddelde();
        }
        bereken = !bereken;
    }
    count = (count +1) % 4;
}

}

void calculateXY(int i, int j, int k){
    long r1 = distances[i];
    long r2 = distances[j];
    long r3 = distances[k];
    long x1 = x[i];
    long x2 = x[j];
    long x3 = x[k];
    long y1 = y[i];
    long y2 = y[j];
    long y3 = y[k];

    long c1 = (((r1*r1)-(r3*r3))-((x1*x1)-(x3*x3))-((y1*y1)-(y3*y3)))/2;
    long c2 = (((r2*r2)-(r3*r3))-((x2*x2)-(x3*x3))-((y2*y2)-(y3*y3)))/2;

    long n = (-((y3-y1)*10000)/(x3-x1));
    long m = ((c1*10000)/(x3-x1));

    long y = ((c2*10000) - m*(x3-x2))/(n*(x3-x2) + (10000*(y3-y2)));
    long x = ((n*y)+m)/10000;

    if(!bereken){
        x_crds[counter] = x;
        y_crds[counter] = y;
    }else{
        x_crds[counter+4] = x;
        y_crds[counter+4] = y;
    }
}

void printGemiddelde(){
    int x_gemiddelde = (x_crds[0]+x_crds[1]+x_crds[2]+x_crds[3]+x_crds[4]+x_crds[5]+x_crds[6]+x_crds[7])/8;
    int y_gemiddelde = (y_crds[0]+y_crds[1]+y_crds[2]+y_crds[3]+y_crds[4]+y_crds[5]+y_crds[6]+y_crds[7])/8;
    printf("De gemiddelde x:%i en y: %i waarden \n\r", x_gemiddelde, y_gemiddelde);
}

```

## B Meetresultaten

(290, 434)		(60, 155)		(72, 377)	
x-coördinaat	y-coördinaat	x-coördinaat	y-coördinaat	x-coördinaat	y-coördinaat
294	434	59	137	75	378
296	445	57	147	77	383
296	441	58	154	76	382
293	434	58	151	76	381
295	436	60	157	75	378
293	434	60	148	74	374
297	439	60	154	74	377
294	432	62	147	75	378
293	449	63	151	76	382
292	449	62	147	77	387
292	449	61	137	76	388
293	449	60	154	78	382
292	439	60	150	74	387
294	439	60	152	76	378
286	433	59	152	74	382
286	433	60	154	76	381
292	436	61	151	77	377
291	435	60	152	75	382
289	429	62	155	77	377
295	436	61	152	75	380
293	444	61	155	76	374
293	435	61	149	73	382
294	451	61	155	75	377
293	433	61	154	74	376
293	433	60	152	77	378
288	440	63	156	78	380
292,9	439,1	60,4	151,8	75,6	379,5
afwijking(290, 434) in cm		afwijking (60, 155) in cm		afwijking(72,377) in cm	
x	2,9	x	0,4	x	3,6
y	5,1	y	3,2	y	2,5
(x,y)	5,87	(x,y)	3,22	(x,y)	4,38